/*
 * Copyright 2011 Ed Clark
 *
 * Licensed under the Apache License, Version 2.0 (the "License");
 * you may not use this file except in compliance with the License.
 * You may obtain a copy of the License at
 *
 *     http://www.apache.org/licenses/LICENSE-2.0
 *
 * Unless required by applicable law or agreed to in writing, software
 * distributed under the License is distributed on an "AS IS" BASIS,
 * WITHOUT WARRANTIES OR CONDITIONS OF ANY KIND, either express or implied.
 * See the License for the specific language governing permissions and
 * limitations under the License.
 */
package meander.StylePanel

import meander.AlignmentType
import meander.PlacementType
import net.miginfocom.swing.MigLayout
import javax.swing.border.TitledBorder

/**
 * GUI tab for alignment and placement style specification
 */
panel( constraints: 'align center', layout: new MigLayout()) {
    panel( border: new TitledBorder( 'Alignment'), layout: new MigLayout()) {
        alignmentGroup = buttonGroup().with {
            AlignmentType.values().toList().each { val ->
                add radioButton( constraints: 'wrap',
                                 text: "Align $val",
                                 selected: bind( source: model.currStyle,
                                                 sourceProperty: 'alignment',
                                                 mutual:true,
                                                 converter: { it == val  },
                                                 reverseConverter: { val }
                                               ),
                                 actionPerformed: { drawExample() },
                               )
            }
        }
    }

    panel( border: new TitledBorder( 'Placement'), layout: new MigLayout( 'flowy, wrap 3')) {
        placementGroup = buttonGroup().with {
            PlacementType.values().toList().each { val ->
                add radioButton( text: "$val",
                                 selected: bind( source: model.currStyle,
                                                 sourceProperty: 'placement',
                                                 mutual:true,
                                                 converter: { it == val  },
                                                 reverseConverter: { val }
                                               ),
                                 actionPerformed: { drawExample() },
                               )
            }
        }
    }
}
